
 #include <Wire.h>
 #include <LiquidCrystal_I2C.h>
 LiquidCrystal_I2C lcd(0x27, 16, 2);
 
 #include <EEPROM.h>
 #include <Talkie.h>
 #include <Vocab_US_Large.h>
 Talkie voice;
 
 /* ── Pin map ─────────────────────────────────────────────── */
 const int CLbarPin  = 2;      // HIGH=C, LOW=L
 const int RmodePin  = 8;      // LOW ⇒ Resistance mode
 const int RelayPin  = 9;      // C2 relay
 const int DecrPin   = 4;      // ↓ CF nudge
 const int OscSigPin = 5;      // oscillator signal
 const int IncrPin   = 6;      // ↑ CF nudge
 const int CalLkPin  = 7;      // jumper enables nudging
 const int SensePin  = A0;     // divider sense for Rx
 
 /* ── Constants ───────────────────────────────────────────── */
 const float FourPiSqrd = 39.4784;
 const float uHmult = 1e6, mHmult = 1e3;
 const float pFmult = 1e12, nFmult = 1e9, uFmult = 1e6;
 const float C2val  = 1.000e-9;      // 1 nF reference
 const float RREF   = 10000.0;       // 10 kΩ reference
 
 float C1val, L1val, CF = 1.0;
 float F1sqrd, F2sqrd;
 long  Fcount, Freq3;  float F3sqrd;
 
 /* ── 1-s polling frequency counter ───────────────────────── */
 void GetFrequency() {
   uint32_t cnt = 0; uint8_t last = PIND & _BV(PD5);
   uint32_t t0 = micros();
   while (micros() - t0 < 1000000UL) {
     uint8_t now = PIND & _BV(PD5);
     if (now && !last) cnt++;
     last = now;
   }
   Fcount = cnt;
 }
 
 /* ── Measure resistance with correct formula --------------- */
 float MeasureRx() {
   uint32_t sum = 0;
   for (uint8_t i = 0; i < 32; i++) sum += analogRead(SensePin);
   float Vout = (sum / 32.0f) * (5.0f / 1023.0f);
   if (Vout < 0.01f || Vout > 4.99f) return -1.0f;      // open/short
   return RREF * (Vout / (5.0f - Vout));                // ohms
 }
 
 /* ── Talkie helpers (digits + units) ----------------------- */
 void speakDigit(uint8_t d){
   static const uint16_t w[]={
     sp2_ZERO,sp2_ONE,sp2_TWO,sp2_THREE,sp2_FOUR,
     sp2_FIVE,sp2_SIX,sp2_SEVEN,sp2_EIGHT,sp2_NINE};
   voice.say(w[d]);
 }

 void speakNumber(float v,uint8_t dec=2){
   int whole = int(v);
   if (!whole) voice.say(sp2_ZERO);
   else {
     char buf[10]; itoa(whole, buf, 10);
     for(char *p = buf; *p; p++) speakDigit(*p - '0');
   }
   if (dec) {
     voice.say(sp2_POINT);
     int frac = int(round((v - whole) * pow(10, dec)));
     char buf2[8]; itoa(frac, buf2, 10);
     for(char *p = buf2; *p; p++) speakDigit(*p - '0');
   }
 }
 
 void speakCap(float C){
   if (C < 1e-9) {                         // pF
     speakNumber(C * pFmult, 2); voice.say(sp2_PICO);
   }
   else if (C < 1e-6) {                    // nF → announce as pF ×1000
     speakNumber(C * nFmult * 1e3, 2); voice.say(sp2_PICO);
   }
   else {                                  // µF
     speakNumber(C * uFmult, 2); voice.say(sp2_MICRO);
   }
   voice.say(sp2_FARAD);
 }
 
 void speakInd(float L){
   if (L < 1e-3) { speakNumber(L * uHmult, 2); voice.say(sp2_MICRO); }
   else if (L < 0.15) { speakNumber(L * mHmult, 2); voice.say(sp2_MILLI); }
   else { voice.say(sp2_OVER); voice.say(sp2_RANGE); return; }
   voice.say(sp2_HENRY);
 }
 
 /* ------------- Resistance speech  -------------- */
 void speakRes(float R){
   if (R < 1e6) {                           // say everything <1 MΩ in ohms
     speakNumber(R, 0); voice.say(sp2_OHMS);
   } else {
     speakNumber(R / 1e6, 2); voice.say(sp2_MEGA); voice.say(sp2_OHMS);
   }
 }
 
 /* ── setup -------------------------------------------------- */
 void setup() {
   pinMode(CLbarPin,INPUT_PULLUP); pinMode(RmodePin,INPUT_PULLUP);
   pinMode(RelayPin,OUTPUT);       pinMode(DecrPin,INPUT_PULLUP);
   pinMode(OscSigPin,INPUT);       pinMode(IncrPin,INPUT_PULLUP);
   pinMode(CalLkPin,INPUT_PULLUP); pinMode(SensePin,INPUT);
 
   lcd.init(); lcd.backlight();
   lcd.setCursor(2,0); lcd.print("Talking Meter");
   delay(1200); lcd.clear();
 
   /* auto-calibration (S1 in C position) */
   digitalWrite(RelayPin,LOW);  GetFrequency(); long F1=Fcount;
   digitalWrite(RelayPin,HIGH); delay(20);      GetFrequency(); long F2=Fcount;
   digitalWrite(RelayPin,LOW);

   F1sqrd = sq(float(F1)); F2sqrd = sq(float(F2));
   C1val  = C2val * (F2sqrd / (F1sqrd - F2sqrd));
   L1val  = 1.0 / (FourPiSqrd * F1sqrd * C1val);
 }
 
 /* ── main loop --------------------------------------------- */
 void loop() {
 
   bool modeR = !digitalRead(RmodePin);   // D8 grounded?
   bool modeC =  digitalRead(CLbarPin);   // S1 in C?
   bool modeL = !modeC && !modeR;
 
   lcd.clear();
 
   /* ---------- RESISTANCE (top priority) ------------------ */
   if (modeR) {
     float Rx = MeasureRx();
     if (Rx < 0) { lcd.print("Open/Short"); voice.say(sp2_OVER); 
voice.say(sp2_RANGE); }
     else {
       if (Rx < 1e6) {
         lcd.print("Rx = "); lcd.print(Rx, 0); lcd.print(" Ohm");
       } else {
         lcd.print("Rx = "); lcd.print(Rx / 1e6, 3); lcd.print(" MOhm");
       }
       speakRes(Rx);
     }
     delay(1500);
     return;
   }
 
   /* ---------- C & L share frequency ---------------------- */
   GetFrequency();
   if (Fcount == 0) { lcd.print("Over Range"); voice.say(sp2_OVER); 
voice.say(sp2_RANGE); delay(1500); return; }
   Freq3  = Fcount; F3sqrd = sq(float(Freq3));
 
   /* ----------------- CAPACITANCE ------------------------- */
   if (modeC) {
     float Cx = C1val * CF * (F1sqrd / F3sqrd - 1.0);
     if      (Cx < 1e-9) { lcd.print("Cx = "); lcd.print(Cx * pFmult, 3); lcd.print(" 
pF"); }
     else if (Cx < 1e-6) { lcd.print("Cx = "); lcd.print(Cx * nFmult, 3); lcd.print(" 
nF"); }
     else                { lcd.print("Cx = "); lcd.print(Cx * uFmult, 3); lcd.print(" 
uF"); }
     speakCap(Cx);
   }
   /* ----------------- INDUCTANCE -------------------------- */
   else { // modeL
     float Lx = L1val * CF * (F1sqrd / F3sqrd - 1.0);
     if      (Lx < 1e-3) { lcd.print("Lx = "); lcd.print(Lx * uHmult, 3); lcd.print(" 
uH"); }
     else if (Lx < 0.15) { lcd.print("Lx = "); lcd.print(Lx * mHmult, 3); lcd.print(" 
mH"); }
     else { lcd.print("Over Range"); voice.say(sp2_OVER); voice.say(sp2_RANGE); 
delay(1500); return; }
     speakInd(Lx);
   }

   /* bottom line shows frequency */
   lcd.setCursor(0,1); lcd.print("F = "); lcd.print(Freq3); lcd.print(" Hz");
 
   /* optional CF nudging */
   if (!digitalRead(CalLkPin)) {
     if (!digitalRead(IncrPin)) { CF *= 1.005; EEPROM.put(0, CF); }
     if (!digitalRead(DecrPin)) { CF *= 0.995; EEPROM.put(0, CF); }
   }
   delay(1500);
 }
 
 /* end of file */

